

Paper 083-2012

Convert Your Old Plots and Charts to New SG Plots and Charts: Here's How

Gabe Cano, Altarum Institute, San Antonio, TX, USA

ABSTRACT

It is definitely in your best interest to move your old SAS/GRAPH® plots and charts to the new SG graphics suite. What was once a timeless exercise in producing quality graphs has now been replaced by a rich graphics suite environment. The SG graphics suite enables you to customize plots and charts with better context-driven graphics options. With a little information, you can quickly get started building or rebuilding your graph plots and charts. SAS® continues to grow its Base SAS® product into a business-level application that can better serve analysts and business users alike. This presentation shows two examples of how to move your old SAS/GRAPH plots and charts to the new SG graphics suite.

KEYWORDS

GPLOT, GCHART, SGPLOT, SGPANEL, BAND

THE SG PROCEDURES

The SG procedures were first introduced in SAS® 9.2 Phase 1. Since their release the SG procedures have gained ground to becoming the new standard for SAS. This suite of graphical applications has become the new wave in providing business solutions to your SAS graphics applications. Whether you are trying to plot just a few points or rendering a complex 3 dimensional scatter plot the time needed to get to a solution is the same ... not much!

THE FORMER STANDARD – PROC GPLOT, GCHART, ETC.

Not so long ago PROC GPLOT was the best way to visualize your data. There is also GCHART, GBARLINE, GCONTOUR, GKPI, GMAP and many other SAS/GRAPH PROCs to get you where you need to be. With just a few steps you could quickly see a sufficient graphical representation of your data. With more time and investment you could create graphics that could warm the heart of even the toughest of clients or bosses. You could incorporate overlaying and replaying of your SAS/GRAPH PROCs to render some pretty amazing pictures. If you were looking for pinpoint accuracy for placing objects on the plot you could use the DSGI and/or ANNOTATE facilities. For dashboard style reports you could use ANNOTATE, GREPLAY and others.


But now there's a better way.

THE NEW STANDARD – PROC SGPLOT

The quick to answer your graphical rendering tasks is here. The SG procedures offer quick business level solutions with much fewer steps. Yes, the system is relatively new so you have to orient yourself in how to find the components that fit your data and application. No, the SG procedures may not completely answer the question you need answers for. At the moment the SG procedures are still heavily in development and we are going to see more and more features appearing from release to release.

This paper intends to help you get started to find those solutions faster. Here's where to start. See FLOWCHART 1.

GRAPHIC 1 – Where to Start with the SG Procedures.



Look at the SGPLOT procedure. This PROC has almost every basic graph facility you will need to get started. From the most basic vector plots to the most complicated splines the SGPLOT procedure offers you the machinery to get you started. Here is the statement list taken from the support.sas.com website for SGPLOT.

The SGPLOT Procedure Syntax:

```

PROC SGPLOT Statement
BAND Statement
DENSITY Statement
DOT Statement
ELLIPSE Statement
HBAR Statement
HBOX Statement
HISTOGRAM Statement
HLINE Statement
INSET Statement
KEYLEGEND Statement
LOESS Statement
NEEDLE Statement
PBSPLINE Statement
REFLINE Statement
REG Statement
SCATTER Statement
SERIES Statement
STEP Statement
VBAR Statement
VBOX Statement
VECTOR Statement
VLINE Statement
XAXIS, X2AXIS, YAXIS, Y2AXIS Statements
  
```

Not all is perfect with the SG procedures. Here is a short list of some challenges you may have to deal with for building your data graphics.

- GOPTIONS do not work with SGPLOT procedures. You may have to customize your output (fonts, height, style, etc.) using the ODS facility line attribute options.

- The SG procedures cannot cycle through symbols and colors the way you can with the SYMBOLi, PATTERNi, etc. statements. You may have to build your own template for colors, symbols, etc.
- Not all the features available in one statement are available in another. For example the FILLATTRS used to fill in the areas under a curve is only available with the BAND, ELLIPSE, HBOX/VBOX, & HBAR/VBAR statements.
- For a more complete list of statements available in the SG procedures refer to “Graphing Made Easy with SG Procedures.” Hands-on workshop, SAS Global Forum 2011 Conference by Slaughter & Delwiche.

Do not fear. The tradeoff for using the SG procedures is quite benign. Let’s look at three basic examples.

Example 1 – Moving from PROC GPLOT to PROC SGPLOT.

Example 2 – Moving from PROC GCHART to PROC SGPLOT.

Example 3 – Moving from PROC GPLOT & ODS TAGSETS.HTMLPANEL to PROC SGPANEL.

EXAMPLE 1 – MOVING FROM PROC GPLOT TO PROC SGPLOT

The following example was taken from SUPPORT.SAS.COM Samples & Notes PROC GPLOT website: Sample 42864 – Scatter plot of groups with an overall regression line. The dataset SASHELP.CLASS is a dataset which comes bundled with SAS 9.2. Let’s look at the records from the SASHELP.CLASS dataset:


DATA 1 – SASHELP.CLASS Data

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83.0
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
9	Jeffrey	M	13	62.5	84.0
10	John	M	12	59.0	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90.0
13	Louise	F	12	56.3	77.0
14	Mary	F	15	66.5	112.0
15	Philip	M	16	72.0	150.0
16	Robert	M	12	64.8	128.0
17	Ronald	M	15	67.0	133.0
18	Thomas	M	11	57.5	85.0
19	William	M	15	66.5	112.0

Suppose we want to graph this dataset in a pseudo BMI context. We want to see the weights by heights plotted for each person in the class. We also want to see females distinguished from males. Here is a quick way to take this simple set of data and produce a graphic representation of it with little or no effort

This is a scatter plot of two groups with an overall regression line added.

OUTPUT 1 – PROC GPLOT Output



Here is the source code that produced OUTPUT 1.

SOURCE CODE 1 – PROC GLOT

```

/* Set the graphics environment */
options reset=all cback=white border htitle=12pt htext=10pt;

/* Define the title */
title1 "Scatter Plot of Groups with an Overall Regression Line";

/* Define the axis characteristics */
axis1 label=("Weight");
axis2 label=(angle=90 "Height") minor=(n=4);

/* Define the symbol characteristics for the scatter plot groups */
symbol1 interpol=none value=dot color=depk;
symbol2 interpol=none value=dot color=vibg;

/* Define the symbol characteristics for the regression line */
symbol3 interpol=r1 value=none color=black;


/* Define the legend options */
legend1 frame label=none repeat=1 value=("Female" "Male");

proc gplot data=sashelp.class;
  plot height*weight=sex / haxis=axis1 vaxis=axis2 legend=legend1;
  plot2 height*weight / noaxis;
run;

```

Now here is the same plot using the SG procedure PROC SGPLOT.

OUTPUT 2 – PROC SGPLOT Output



Here is the source code that produced OUTPUT 2.

SOURCE CODE 2 – PROC SGPLOT

```
proc template;
Define style styles.mystyle;
  Parent=styles.default;
  Style graphdata1 from graphdata1 /
    MarkerSymbol="circlefilled" Color=vibg Contrastcolor=vibg;
  Style graphdata2 from graphdata2 /
    MarkerSymbol="circlefilled" Color=depk Contrastcolor=depk;
end;
run;

proc format;
  value $sexfmt
    'F'='Female'
    'M'='Male'
  ;
run;

/* Define the title */
title "Scatter Plot of Groups with an Overall Regression Line";
```

```
ods listing style=mystyle;

proc sgplot data=sashelp.class;
  format sex $sexfmt.;
  reg X=weight Y=height / markerattrs=(color=white) lineattrs=(color=vib);
  scatter X=weight Y=height / group=sex name='x_y';
  keylegend 'x_y' / border title=' ';
  xaxis values=(50 to 150 by 10);
  yaxis values=(50 to 80 by 10);
run;
```

For the most part these two graphics are comparable. In the case of PROC GPLOT we were able to control the symbols and colors used in GRAPH 1. With a modification to the STYLES template and creating a PROF FORMAT for the legend label we have almost identical output.

EXAMPLE 2 – MOVING FROM PROC GCHART TO PROC SGPLOT

The following example 2 was taken from SUPPORT.SAS.COM Samples & Notes PROC GCHART website: Sample 25512 – Midpoints/stats in horizontal bar chart using PROC GCHART.

The dataset is given here.


DATA 2 – Fitness Data

Obs	age	sex	heart	exer	aero
1	39	F	90	1	13.6
2	28	M	96	1	33.0
3	46	M	84	1	34.2
4	25	F	88	1	32.0
5	.	F	70	1	22.8
6	36	F	80	1	26.2
7	28	F	76	1	23.8
8	23	F	68	1	18.9
9	31	F	84	1	21.6
10	32	F	72	1	15.7
11	30	F	66	2	28.9
12	24	M	82	2	18.7
13	28	M	86	2	36.6
14	30	M	78	2	33.8
15	26	M	74	2	42.7
16	48	F	72	2	22.6
17	28	F	84	2	22.1
18	37	F	78	2	30.3
19	23	M	72	2	38.1
20	37	F	72	2	43.7
21	24	F	74	2	29.9
22	48	F	72	2	28.9
23	22	F	78	2	39.0
24	45	M	60	2	47.8
25	27	M	90	2	43.1
26	26	M	66	2	28.9
27	26	F	84	2	.
28	48	F	66	2	28.9
29	36	F	66	2	33.2
30	24	M	72	3	50.1
31	42	M	60	3	36.7

32	44	F	78	3	21.6
33	31	M	60	3	44.1
34	41	M	76	3	26.7
35	25	F	60	3	36.1
36	24	M	72	3	44.2
37	23	M	54	3	70.6
38	25	M	54	3	41.3
39	23	F	66	3	36.1
40	46	F	54	3	28.9
41	29	M	54	3	44.8
42	.	F	66	4	36.1
43	33	F	56	4	21.3
44	29	F	66	4	38.9
45	27	M	62	4	44.0

Plotting this data using PROC GCHART gives the following graphic.

OUTPUT 3 – PROC GCHART Output



Here is the PROC GCHART code that produced OUTPUT 3.

SOURCE CODE 3 – PROC GCHART

```

/* Set the graphics environment */
goptions reset=all border cback=white htitle=12pt;

title1 'Fitness Program Participants';

/* Define response axis characteristics */

```

```

axis1 label=('Number of People')
      minor=(number=1)
      offset=(0,0);


/* Define legend characteristics */
legend1 label=none value=('Women' 'Men');

/* Chart the values of AGE */
proc gchart data=fitness;
  hbar age / subgroup=sex
        legend=legend1
        autoref
        clipref
        coutline=black
        raxis=axis1;
run;

```

Plotting the same data using SGPLOT we get the following.

OUTPUT 4 – PROC SGPLOT Output



Here is the PROC GGPLOT code that produced OUTPUT 4.

SOURCE CODE 4 – PROC SGPLOT

```

proc format;
  value $sexfmt

```

```

    'F'='Women'
    'M'='Men'
;
run;

data fitness;
set fitness;
format sex $sexfmt.;
if 20 <= age <= 24 then age_midpoint = 1;
else if 25 <= age <= 29 then age_midpoint = 2;
else if 30 <= age <= 34 then age_midpoint = 3;
else if 35 <= age <= 39 then age_midpoint = 4;
else if 40 <= age <= 44 then age_midpoint = 5;
else if 45 <= age <= 49 then age_midpoint = 6;
run;

title 'Fitness Program Participants';

/* Chart the values of AGE */
proc sgplot data=fitness;
  hbar age_midpoint / group=sex datalabel name='m_f';
  keylegend 'm_f' / noborder;
  xaxis label='Number of People' grid values=(0 to 15 by 1);
run;


```

Comparing and contrasting PROC GCHART versus PROC SGPLOT we can see that GCHART may have the edge here or it may not. In order to create an SGPLOT graph that was comparable to GCHART a couple of items had to be added. For example, an age midpoint variable had to be constructed. GCHART automatically created one for us. This is good if that's what you wanted it to do. Also note that GCHART printed a cumulative frequency of the data to the right of the OUTPUT 3 graphic. Again, this is good if that's what you wanted the output to look like.

EXAMPLE 3 – USING ODS.TAGSETS.HTMLPANEL TO CREATE QUICK DASHBOARDS

Finally, let's look at one last example. Suppose we want to plot the vectors given in GRAPHIC 2 for the unit square. (Reference SGF 2011 paper 063-201 for more details.)

GRAPHIC2 – The Unit Square



Recall the data for this example.


DATA 3 – Unit Square Vector Data

Obs	chart	x	y
1	1	0.0	0.0
2	1	0.0	0.0
3	2	0.0	0.0
4	2	1.0	0.0
5	3	0.0	0.0
6	3	1.0	0.2
7	4	0.0	0.0
8	4	1.0	0.4
9	5	0.0	0.0
10	5	1.0	0.6
11	6	0.0	0.0
12	6	1.0	0.8
13	7	0.0	0.0
14	7	1.0	1.0
15	8	0.0	0.0
16	8	0.8	1.0
17	9	0.0	0.0
18	9	0.6	1.0
19	10	0.0	0.0
20	10	0.4	1.0
21	11	0.0	0.0
22	11	0.2	1.0
23	12	0.0	0.0
24	12	0.0	1.0

Note that the highlighted points can be deleted for PROC GPLOT but must be included if using the BAND statement in SGLOT or SGPANEL.

Here is the final output from PROC GPLOT for this dataset.


OUTPUT 5 – PROC GPLOT & ODS.TAGSETS.HTMLPANEL Output



Reference SGF 2011 paper 063-2011 for the code that produced OUTPUT 5.

Now let's look at the output from PROC SGPANEL.

OUTPUT 6 – PROC SGPANEL Output



Here is the code that produce OUTPUT 6.

SOURCE CODE 5 – PROC SGPANEL

```
proc sgpanel data=test1 noautolegend;
  panelby chart / columns=3 rows=4;
  band X=x lower=0 upper=y / fillattrs=(color=salmon);
  rowxaxis display=(nolabel);
  colxaxis display=(nolabel);
run;
```

This is amazing! Less code was used to create sufficiently comparable output.

CONCLUSION

I hope that you have gained some insight in how to approach using the SG procedures. It is to your extreme benefit to follow this path in developing your SAS graph applications. As SAS continues to improve the ODS system and graphics facilities we continue to reap the benefits in our ability to deliver better products.

REFERENCES

Slaughter, Susan, Delwiche, Lora D. 2011. "Graphing Made Easy with SG Procedures." .” *Proceedings of the SAS Global Forum 2011 Conference*. Paper 174-2011.

Robert Allison's website: <http://robslink.com/SAS/Home.htm>.

Swartz, Susan. 2009. "Clinical Trial Reporting Using SAS/GRAPH SG Procedures." .” *Proceedings of the SAS Global Forum 2009 Conference*. Paper 174-2009.

Heath, Dan. 2009. "Secrets of the SG Procedures." *Proceedings of the SAS Global Forum 2009 Conference*. Paper 324-2009.

Example 1 – This PROC GPLOT example was taken from support.sas.com Samples & Notes PROC GPLOT website: Sample 42864 – Scatter plot of groups with an overall regression line.

Reference URL: http://support.sas.com/sassamples/graphgallery/PROC_GPLOT.html

Example 2 – This PROC GCHART example was taken from the support.sas.com Support Samples & Notes PROC GCHART website: Sample 25512 – Midpoints/stats in horizontal bar chart using PROC GCHART.

Reference URL: http://support.sas.com/sassamples/graphgallery/PROC_GCHART.html

Cano, Gabe. 2011. "Using ODS TAGSETS.HTMLPANELS to Create Quick Dashboards." .” *Proceedings of the SAS Global Forum 2011 Conference*. Paper 063-2011.

ACKNOWLEDGEMENTS

I would like to acknowledge and thank Susan Slaughter for presenting her hands-on workshop presented at SGF 2011 – Graphing Made Easy with SG Procedures. This workshop helped me to get familiar with the SG procedures environment and to move forward with their applications.

I would also like to acknowledge and thank Robert Allison for his website: <http://robslink.com/SAS/Home.htm> His tireless efforts to post his numerous examples have helped me to apply SAS better in my work environment.

CONTACT INFORMATION

Gabe Cano
Senior Analyst
Altarum Institute
10001 Reunion Place
Union Square II, Suite #200
San Antonio, TX 78216
gabe.cano@altarum.org
Phone: (210)832-3013
Receptionist: (210)832-3000
FAX: (210)832-3099
<http://www.altarum.org>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.